Product of two Gaussian probabilistic density functions

Luis Pedro Coelho

April 19, 2013

> **What is this?**
> Since there are a few hits of people who download these files directly from a Google results page, instead of from my website, I add here a small explanation:
> These files contain derivations which I often use and, before I wrote them down cleanly, would often make small mistakes on (like leaving off a minus sign or such). They are also automatically checked with **sage**, so that the computer assures that there were no mis-steps in the derivation.
> This is also why you see code in the derivation instead of just the equations. This is run by **sage**.

Let's start by defining a few variables:

```
x = var('x')
mu1 = var('mu1')
mu2 = var('mu2')
```

x will be our variable, μ_1 and μ_2 the means.

```
s1 = var('s1', latex_name=r'\sigma_1^2', domain='positive')
s2 = var('s2', latex_name=r'\sigma_2^2', domain='positive')
N1, N2 = var('N1 N2')
```

We define σ^2 as positive.

```
Z(s) = sqrt(2*pi*s)
N(x,m,s) = 1./Z(s) * exp(- (x-m)^2 /(2*s))

N1 = N(x, mu1, s1)
N2 = N(x, mu2, s2)
```

`N1` is the first normal:

$$\frac{0.500000000000000\,\sqrt{2}e^{\left(-\frac{(\mu_1-x)^2}{2\,\sigma_1^2}\right)}}{\sqrt{\pi\sigma_1^2}}, \tag{1}$$

which is not in canonical form because of sage simplification, but we can recognise the normal distribution.

```
product = N1 * N2
```

The product as a Gaussian is readily obtained by algebraic manipulation:

```
m12 = (mu1*s2+mu2*s1)/(s1+s2)
s12 = s1*s2/(s1+s2)
newnormal = N(x,m12,s12)
```

We have that

$$\mu_{12} = \frac{\mu_1\sigma_2^2 + \mu_2\sigma_1^2}{\sigma_1^2 + \sigma_2^2} \tag{2}$$

$$\sigma_{12}^2 = \frac{\sigma_1^2\sigma_2^2}{\sigma_1^2 + \sigma_2^2} \tag{3}$$

However, to obtain the full product, we need to add several normalization factors:

```
direct = 1./Z(s1)*1./Z(s2)*\
        exp(- (mu1-mu2)^2/2/(s1+s2))*Z(s12)*newnormal
```

Which is

$$\frac{0.500000000000000\,e^{\left(-\frac{\left(x-\frac{\mu_1\sigma_2^2+\mu_2\sigma_1^2}{\sigma_1^2+\sigma_2^2}\right)^2\left(\sigma_1^2+\sigma_2^2\right)}{2\,\sigma_1^2\sigma_2^2}-\frac{(\mu_1-\mu_2)^2}{2\,\left(\sigma_1^2+\sigma_2^2\right)}\right)}}{\sqrt{\pi\sigma_1^2}\sqrt{\pi\sigma_2^2}}. \tag{4}$$

Let's finally check that this is correct, by checking the value of the ratio:

```
final = (product/direct).full_simplify()
```

And the output is 1!